

LISTA SELECAO – VISUALG

// Modificar o ex1A para dar 3 respostas: **intervalo/abaixo/acima**

ALGORITMO "ex1Amodificado"

```
VAR
    numero: INTEIRO
INICIO
    LEIA (numero)
    SE (numero >= 20) E (numero <= 90) ENTÃO
        ESCREVA ("Numero no intervalo [20,90]")
    SENÃO
        SE numero < 20 ENTÃO
            ESCREVA ("Numero abaixo do intervalo [20,90]")
        SENÃO
            ESCREVA ("Numero acima do intervalo [20,90]")
        FIMSE
    FIMSE
FIMALGORITMO
```

// Modificar para dar respostas em outra ordem (**abaixo/intervalo/acima**)

// gera menos testes

ALGORITMO "ex1Amelhorado"

```
VAR
    numero: INTEIRO
INICIO
    LEIA (numero)
    SE numero < 20 ENTÃO
        ESCREVA ("Numero abaixo do intervalo [20,90]")
    SENÃO
        SE numero <= 90 ENTÃO
            ESCREVA ("Numero no intervalo [20,90]")
        SENÃO
            ESCREVA ("Numero acima do intervalo [20,90]")
        FIMSE
    FIMSE
FIMALGORITMO
```

// Modificar o ex2A para dar 3 respostas: max é **a/b/c**

ALGORITMO "ex2Amodificado"

```
VAR
    a, b, c, max: INTEIRO
INICIO
    LEIA (a, b, c)
    SE (a > b) E (a > c) ENTÃO
        max <- a
    SENÃO
        SE b > c ENTÃO
            max <- b
        SENÃO
            max <- c
        FIMSE
    FIMSE
    ESCREVA ("Maior entre ", a, " ", b, " ", c, ": ", max)
FIMALGORITMO
```

3. Escreva um algoritmo que leia três valores **a**, **b** e **c**, e calcule e escreva a média ponderada com pesos de 5 para o maior dos 3 valores e 2.5 para os outros 2 valores.

ALGORITMO "ex3"

```
VAR
    a, b, c, media: REAL
INICIO
    ESCREVA ("Informe os valores")
    LEIA (a,b,c)
    SE (a>b) E (a>c) ENTÃO
        media <- (a * 5 + b * 2.5 + c * 2.5) / 10
    SENÃO // nota a eh descartada pra comparacao
        SE b>c ENTÃO
            media <- (a * 2.5 + b * 5 + c * 2.5) / 10
        SENÃO // sobra o maior sendo c
            media <- (a * 2.5 + b * 2.5 + c * 5) / 10
        FIMSE
    FIMSE
    ESCREVAL ("Media: ", media)
FIMALGORITMO
```

4. Escreva um algoritmo que leia 2 valores inteiros, a e b , e escreva-os juntamente com a mensagem 'são múltiplos' ou 'não são múltiplos', conforme o caso.

ALGORITMO "ex4"

```
VAR
    a, b: INTEIRO
INICIO
    ESCREVA ("Informe os valores")
    LEIA (a,b)
    SE (a MOD b = 0) OU (b MOD a = 0) ENTÃO // valores iguais: considerar múltiplos
        ESCREVAL (a, " ", b, "MULTIPLOS")
    SENÃO
        ESCREVAL (a, " ", b, "NAO SAO MULTIPLOS")
    FIMSE
FIMALGORITMO
```

5. Escreva um algoritmo que leia 4 números inteiros (opção, n_1 , n_2 , n_3) e escreva-os da seguinte maneira:

- se opção = 1 escreva o número n_1
- se opção = 2 escreva o número n_2
- se opção = 3 escreva o número n_3
- se opção for outro valor qualquer escreva opção inválida

ALGORITMO "ex5"

```
VAR
    opcao, n1, n2, n3: INTEIRO
INICIO
    ESCREVA ("Informe os valores")
    LEIA (opcao, n1, n2, n3)
    ESCOLHA (opcao)
        CASO 1
            ESCREVAL (n1)
        CASO 2
            ESCREVAL (n2)
        CASO 3
            ESCREVAL (n3)
        OUTROCASO
            ESCREVAL ("OPCAO INVALIDA!!")
    FIMESCOLHA
FIMALGORITMO
```

6. Escreva um algoritmo que leia 4 valores (opção, a , b , c), onde opção é um valor inteiro e positivo e a , b , c são quaisquer valores reais. Escreva os valores lidos da seguinte maneira:

- se opção = 1 \Rightarrow escreva os 3 valores a , b , c em ordem crescente
- se opção = 2 \Rightarrow escreva os 3 valores a , b , c em ordem decrescente
- se opção = 3 \Rightarrow escreva os 3 valores de forma que o maior valor entre a , b , c fica entre os outros 2.

ALGORITMO "ex6"

```
VAR
    a, b, c, max, meio, min: REAL
    opcao: INTEIRO
INICIO
    ESCREVA ("Informe os valores")
    LEIA (opcao, a,b,c)
    SE (a>b) E (a>c) ENTÃO // 1a tarefa: resolver quem eh max, med, min
        max <- a // esta tarefa nao depende da var opcao
        SE b>c ENTÃO
            meio<- b
            min<- c
        SENÃO
            meio<- c
            min<- b
        FIMSE
    SENÃO
        SE b>c ENTÃO
            max <- b
            SE a>c ENTÃO
                meio<- a
                min<- c
            SENÃO
                meio<- c
                min<- a
            FIMSE
        SENÃO
            max <- c
            SE a>b ENTÃO
                meio<- a
                min<- b
            SENÃO
                meio<- b
```

```

        min<- a
    FIMSE
FIMSE
ESCOLHA (opcao) // 2a tarefa: como escrever (depende de opcao)
    CASO 1
        ESCREVAL (min, " ", meio, " ", max)
    CASO 2
        ESCREVAL (max, " ", meio, " ", min)
    CASO 3
        ESCREVAL (meio, " ", max, " ", min)
FIMESCOLHA
FIMALGORITMO

```

7. Faça um algoritmo que leia 3 valores x, y, z, e verifique se podem ser os comprimentos dos lados de um triângulo. Em caso afirmativo, verifique se é 'triângulo **equilátero**', 'triângulo **isósceles**' ou 'triângulo **escaleno**'. Em caso negativo, escreva uma mensagem: 'os valores lidos não formam um triângulo'. Considere que:

– o comprimento de **cada lado** de um triângulo é **menor que a soma dos comprimentos dos outros lados**

– um triângulo **equilátero** tem três lados iguais

– um triângulo **isósceles** tem dois lados iguais e um diferente

– um triângulo **escaleno** tem três lados diferentes.

ALGORITMO "ex7"

```

VAR
    x, y, z: REAL
INICIO
    ESCREVA ("Informe os lados do triangulo")
    LEIA (x, y, z)
    SE (x >= y+z) OU (y >= x+z) OU (z >= x+y) ENTAO
        ESCREVAL ("Os valores lidos não formam um triangulo")
    SENAO
        SE (x = y) E (y = z) ENTAO // nao precisa testar se x igual a z
            ESCREVAL ("Os valores lidos formam um triangulo equilatero")
        SENAO
            SE (x = y) OU (x = z) OU (y = z) ENTAO
                ESCREVAL ("Os valores lidos formam um triangulo isosceles")
            SENAO
                ESCREVAL ("Os valores lidos formam um triangulo escaleno")
        FIMSE
    FIMSE
FIMSE
FIMALGORITMO

```

ALGORITMO "ex7-versao2"

```

VAR
    x, y, z: REAL
INICIO
    ESCREVA ("Informe os lados do triangulo")
    LEIA (x, y, z)
    SE (x < y+z) E (y < x+z) E (z < x+y) ENTAO
        SE (x = y) E (y = z) ENTAO // nao precisa testar se x igual a z
            ESCREVAL ("Os valores lidos formam um triangulo equilatero")
        SENAO
            SE (x = y) OU (x = z) OU (y = z) ENTAO
                ESCREVAL ("Os valores lidos formam um triangulo isosceles")
            SENAO
                SE x<>y E x<>z E y<>z ENTAO // TST Q NÃO DEVE SER FEITO, TST DESNECESSÁRIO!!
                ESCREVAL ("Os valores lidos formam um triangulo escaleno")
            FIMSE
        FIMSE
    FIMSE
FIMSE
SENÃO
    ESCREVAL ("Os valores lidos não formam um triangulo")
FIMSE
FIMALGORITMO

```

8. Faça um algoritmo para calcular as raízes reais de uma equação quadrática: $ax^2 + bx + c = 0$. Uma equação quadrática só tem raiz reais se $(b^2 - 4ac)$ for maior ou igual a zero. O algoritmo deve ser capaz de tratar os casos em que há duas raízes distintas, uma raiz real ou nenhuma raiz.

ALGORITMO "ex8"

```

VAR
    a, b, c, x, delta, raiz1, raiz2: REAL
INICIO
    ESCREVA ("Informe os valores da equação: a, b, c e x")
    LEIA (a, b, c, x)
    delta <- b^2 - 4*a*c

```

```

SE delta < 0 ENTAO
    ESCREVAL ("Esta equacao nao tem raizes reais")
SENAO
    SE delta = 0 ENTAO
        raiz1 <- (-b)/(2*a)
        ESCREVAL ("A equacao tem uma unica raiz: ", raiz1)
    SENAO
        raiz1 <- (-b + RAIZQ(delta))/(2*a)
        raiz2 <- (-b - RAIZQ(delta))/(2*a)
        ESCREVAL ("A equacao tem uma raizes: ", raiz1, " e ", raiz2)
    FIMSE
FIMSE
FIMALGORITMO

```

9. Uma empresa decidiu conceder um aumento de salário a seus funcionários de acordo com a tabela:

Em R\$	Índice de aumento
salário ≤ 400.00	15%
400.00 < salário ≤ 700.00	12%
700.00 < salário ≤ 1000.00	10%
1000.00 < salário ≤ 1500.00	7%
1500.00 < salário ≤ 2000.00	4%
salário > 2000.00	sem aumento

Faça um algoritmo que leia o salário atual de um funcionário e escreva o índice de aumento e o valor do salário corrigido.

ALGORITMO "ex9"

```

VAR
    indiceAumento: INTEIRO
    salarioAtual, salarioCorrigido: REAL
INICIO
    ESCREVA ("Informe salario atual")
    LEIA (salarioAtual)
    SE salarioAtual <= 400.00 ENTAO
        indiceAumento <- 15
    SENAO // 400.01->
        SE salarioAtual <= 700.00 ENTAO
            indiceAumento <- 12
        SENAO
            SE salarioAtual <= 1000.00 ENTAO
                indiceAumento <- 10
            SENAO
                SE salarioAtual <= 1500.00 ENTAO
                    indiceAumento <- 7
                SENAO
                    SE salarioAtual <= 2000.00 ENTAO
                        indiceAumento <- 4
                    SENAO
                        indiceAumento <- 0
                FIMSE
            FIMSE
        FIMSE
    FIMSE
    salarioCorrigido <- salarioAtual + indiceAumento/100 * salarioAtual
    ESCREVAL ("Índice de aumento: ", indiceAumento, "% - Salario Corrigido: R$", salarioCorrigido)
FIMALGORITMO

```

10. A 3NET decidiu dar a seus funcionários uma gratificação de Páscoa. Para cada empregado da empresa tem-se nome, número de horas extras trabalhadas e número de faltas. Para calcular o prêmio, subtrai-se dois terços das horas que o empregado não trabalhou de suas horas extras e atribui-se a gratificação segundo a tabela ao lado.

ht = horas extras - 2/3 * faltas	Gratificação
ht > 40 horas	1000,00
30 < ht ≤ 40	800,00
20 < ht ≤ 30	600,00
10 < ht ≤ 20	400,00
ht ≤ 10	200,00

ALGORITMO "ex10"

```

VAR
    nome: CARACTERE
    nfaltas: INTEIRO
    hextra, ht, grat: REAL
INICIO
    ESCREVA ("Informe Nome, numero de horas extras e numero de faltas")
    LEIA (nome, hextra, nfaltas)
    ht <- hextra - 2/3 * nfaltas // em C++ 2/3 dara 0 (int/int), trocar para 2/3.0
    SE ht > 40 ENTAO
        grat <- 1000.00
    SENAO
        SE ht > 30 ENTAO
            grat <- 800.00

```

```

SENÃO
  SE ht>20 ENTÃO
    grat <- 600.00
  SENÃO
    SE ht>10 ENTÃO
      grat <- 400.00
    SENÃO
      grat <- 200.00
  FIMSE
FIMSE
FIMSE
  ESCREVAL ("Gratificacao de ", nome, " = R$", grat)
FIMALGORITMO

```

11. Escreva um algoritmo que leia a hora de início de um jogo e a hora final do jogo (considerando apenas horas inteiras) e calcule a duração do jogo em horas, sabendo-se que o tempo máximo de duração do jogo é de 24 horas e que o jogo pode iniciar em um dia e terminar no dia seguinte.

ALGORITMO "ex11"

```

VAR
  horainicial, horafinal: INTEIRO
INICIO
  ESCREVA ("Informe as horas de inicio e fim do jogo")
  LEIA (horainicial, horafinal)
  SE horainicial < horafinal ENTÃO // jogo no mesmo dia
    ESCREVAL ("Jogo durou ", horafinal - horainicial, "h")
  SENÃO
    ESCREVAL ("Jogo durou ", 24 - horainicial + horafinal, "h") //p.ex. 24h - 21h + 5h
  FIMSE
FIMALGORITMO

```

12. Escreva um algoritmo que leia a matrícula, as 3 notas obtidas por um aluno nas 3 verificações parciais e a média dos exercícios (ME) que fazem parte da avaliação. Calcule a média de aproveitamento, usando a fórmula:

$$\text{Média de aproveitamento} = \frac{VP1A + VP1B * 2 + VP2 * 3 + ME}{7}$$

A atribuição dos conceitos obedece à tabela abaixo:

Média de aproveitamento	conceito
≥ 9.0	A
≥ 7.5 e < 9.0	B
≥ 6.0 e < 7.5	C
≥ 4.0 e < 6.0	D
< 4.0	E

O algoritmo deve escrever a matrícula do aluno, as notas das verificações, a média dos exercícios, a média de aproveitamento, o conceito correspondente e a mensagem 'aprovado' caso o conceito seja A, B ou C, e 'reprovado' caso o conceito seja D ou E.

ALGORITMO "ex12"

```

VAR
  vpla, vplb, vp2, me, media: REAL
  conceito, matr, sit: CARACTERE
INICIO
  ESCREVA ("Informe os dados")
  LEIA (vpla, vplb, vp2, me)
  media <- (vpla + vplb * 2 + vp2 * 3 + me) / 7
  SE media >= 9.0 ENTÃO
    conceito <- "A"
  SENÃO
    SE media >= 7.5 ENTÃO
      conceito <- "B"
    SENÃO
      SE media >= 6.0 ENTÃO
        conceito <- "C"
      SENÃO
        SE media >= 4.0 ENTÃO
          conceito <- "D"
        SENÃO
          conceito <- "E"
        FIMSE
      FIMSE
    FIMSE
  FIMSE
  ESCOLHA (conceito)
    CASO "A", "B", "C"
      sit <- "Aprovado"
    OUTROCASO
      sit <- "Reprovado"
  FIMESCOLHA

```

```

    ESCRVAL ("Aluno ", matr, "obteve ", ma, " - ", conceito, " - ", sit)
FIMALGORITMO

```

13. O departamento do meio ambiente mantém 3 listas (lista A, lista B e lista C) de indústrias conhecidas por serem altamente poluentes da atmosfera. Os resultados de várias medidas são combinados para formar o que é chamado de "índice de poluição". Isso é controlado regularmente. Normalmente os valores variam entre 0.05 e 0.25. Se o valor atingir 0.30, as indústrias da lista A serão chamadas a suspender as operações até que os valores retornem ao intervalo normal. Se o índice atingir 0.40, as indústrias da lista B serão notificadas também. Se o índice exceder 0.50, indústrias de todas as 3 listas serão avisadas para suspenderem as atividades. Faça um algoritmo para ler o índice de poluição e indicar as notificações apropriadas.

ALGORITMO "ex13"

```

VAR
    poluicao: REAL
INICIO
    ESCRVA ("Informe o Indice de Poluicao")
    LEIA (poluicao)
    SE (poluicao >= 0.05) E (poluicao < 0.30) ENTAO //0.30 p/nao ficar valor sem resp
        ESCRVAL ("Indice de poluicao normal!!")
    SENAO // hierarquico: SE-FIMSE
        SE poluicao > 0.5 ENTAO // exceder 50
            ESCRVAL ("Indice atingiu 50%! Suspender operacoes na categoria C")
        FIMSE
        SE poluicao >= 0.40 ENTAO // atingir 40
            ESCRVAL ("Indice atingiu 40%! Suspender operacoes na categoria B")
        FIMSE
        SE poluicao >= 0.30 ENTAO // atingir 30
            ESCRVAL ("Indice atingiu 30%! Suspender operacoes na categoria A")
        FIMSE
    FIMSE
FIMALGORITMO

```

14. Uma lanchonete *fast food* apresenta a seguinte relação de produtos:

Código	Descrição	Preço (r\$)	
1	Hamburger	14,50	Alimentos
2	Chessburger	15,50	
3	Cachorro-Quente	10,00	
4	Sanduíche Natural	13,50	
5	Refrigerante	5,00	Bebidas
6	Suco de Laranja	6,00	
7	Milk shake	10,50	
8	Sundae	13,00	Sobremesas
9	Casquinha	10,00	

Cada cliente sempre deve pedir um item de alimentação (01 a 04), uma bebida (05 ou 06) e uma sobremesa (07 a 09). Escreva um algoritmo que leia os 3 códigos do pedido do cliente e Informe a descrição/preço de cada item e o preço final a pagar. Caso ocorra algum problema (código inválido ou repetição de algum item), exibir apenas uma mensagem de erro.

ALGORITMO "ex14"

```

VAR
    codA, codB, codS: INTEIRO
    total: REAL
INICIO
    ESCRVA ("Informe os codigos de alimento, bebida e sobremesa")
    LEIA (codA, codB, codS)
    SE (codA<1) OU (codA>4) ENTAO // testando se pedido eh invalido
        ESCRVAL ("Alimento Invalido - saindo... ")
    SENAO
        SE (codB<5) OU (codB>6) ENTAO
            ESCRVAL ("Bebida Invalida - saindo... ")
        SENAO
            SE (codS<7) OU (codS>9) ENTAO
                ESCRVAL ("Sobremesa Invalida - saindo... ")
            SENAO
                ESCRVAL ("Descricao do pedido: ") // aqui o pedido estah correto
                ESCOLHA (codA)
                CASO 1
                    total <- 14.5
                    ESCRVAL ("hamburger - 14,50")
                CASO 2
                    total <- 15.5
                    ESCRVAL ("chessburger - 15,50")
                CASO 3
                    total <- 14.0
                    ESCRVAL ("hot dog - 14,00")
                CASO 4
                    total <-13.5
                    ESCRVAL ("sanduiche - 13,50")
                FIMESCOLHA
    FIMSE
FIMALGORITMO

```

```

ESCOLHA (codB)
CASO 5
    total <- total + 5.0
    ESCREVAL ("refrigerante - 5,00")
CASO 6
    total <- total + 6.0
    ESCREVAL ("suco de laranja - 6,00")
FIMESCOLHA
ESCOLHA (codS)
CASO 7
    total <- total + 10.5
    ESCREVAL ("milk shake - 10,50")
CASO 8
    total <- total + 13.0
    ESCREVAL ("sundae - 13,00")
CASO 9
    total <- total + 10.0
    ESCREVAL ("casquinha - 10,00")
FIMESCOLHA
    ESCREVAL ("Total a pagar R$", total)

```

FIMSE

FIMSE

FIMSE

FIMALGORITMO

15. Faça um algoritmo leia um caracter indicando uma opção:

- se opção = 'T' calcular a área de um triângulo de base b e altura h

- se opção = 'R' calcular a área de um **retângulo** de base b e altura h

- se opção = 'Q' calcular a área de um quadrado de lado l

ALGORITMO "ex15"

```

VAR
    figura: CARACTERE
    area, med1, med2: REAL
INICIO
    ESCREVA ("Informe o codigo da figura (T,R,Q): ")
    LEIA (figura)
    ESCOLHA (figura)
    CASO "T", "R":
        ESCREVA ("Informe base e altura: ")
        LEIA (med1, med2)
        area <- med1 * med2
        SE figura = "T" ENTÃO
            area <- area / 2
        FIMSE
    CASO "Q":
        ESCREVA ("Informe lado: ")
        LEIA (med1)
        area <- med1 ^ 2
    FIMESCOLHA
    ESCREVAL ("Area: ", area)
FIMALGORITMO

```

ALGORITMO "ex15-versao2"

```

//...
ESCOLHA (figura)
CASO "T":
    ESCREVA ("Informe base e altura: ")
    LEIA (med1, med2)
    area <- med1 * med2 / 2
CASO "R":
    ESCREVA ("Informe base e altura: ")
    LEIA (med1, med2)
    area <- med1 * med2
CASO "Q":
    ESCREVA ("Informe lado: ")
    LEIA (med1)
    area <- med1 ^ 2
FIMESCOLHA

```

16. Faça um algoritmo que leia o sexo e altura de uma pessoa e calcule seu peso ideal, considerando que para homens o peso ideal é igual a $(72,7 * altura) - 58$ e para mulheres o peso ideal é igual a $(62,1 * altura) - 44,7$.

ALGORITMO "ex16"

```

VAR
    sexo: CARACTERE
    altura, pesoIdeal: REAL
INICIO
    ESCREVA ("Informe sexo, altura e peso:")
    LEIA (sexo, altura)
    ESCOLHA (MAIUSC(sexo)) // poderia resolver com se-senao
    CASO "F":
        pesoIdeal <- 62.1 * altura - 44.7
    CASO "M": // poderia usar OUTROCASO: se o dado de entrada estivesse validado
        pesoIdeal <- 72.7 * altura - 58
    FIMESCOLHA
    ESCREVAL ("Seu peso ideal eh ", pesoIdeal)
FIMALGORITMO

```

17.O cardápio de uma lanchonete é dado pela tabela abaixo:

Código Produtos	Preço Unitário (r\$)
100 Cachorro-quente + refrigerante	13,00
101 Misto quente + refrigerante	12,50
102 Misto frio + refrigerante	12,00
103 Queijo quente + refrigerante	12,25

Durante essa semana, a lanchonete está em promoção: para qualquer lanche de um mesmo tipo adquirido, **leve 5 e pague 4**. Faça um algoritmo que leia o código do pedido e o número de itens de um mesmo tipo a ser adquirido por um consumidor e escreva o valor a pagar. **Caso o código do pedido não seja válido, escreva apenas uma mensagem de erro.** O consumidor só pode pedir itens de um mesmo tipo.

ALGORITMO "ex17"

```
VAR
    cod, qtde, qtddesconto: INTEIRO
    total, preco: REAL
INICIO
    ESCRIVA ("Informe o codigo do pedido e a quantidade")
    LEIA (cod, qtde)
    SE (cod<100) OU (cod>103) ENTAO
        ESCRIVAL ("Pedido Invalido - saindo...")
    SENAO
        ESCOLHA (cod)
        CASO 100
            preco <- 13
        CASO 101
            preco <- 12.5
        CASO 102
            preco <- 12.0
        CASO 103
            preco <- 12.25
        FIMESCOLHA
        qtddesc <- qtde DIV 5 // cada 5 pg 4(1 desc), p/10 desc 2, ...
        total <- (qtde - qtddesc) * preco
        ESCRIVAL ("A pagar R$", total)
    FIMSE
FIMALGORITMO
```

18. Faça um algoritmo que leia um número inteiro representando um determinado mês do ano e escreva o mês por extenso. Para valores menores do que 1 ou maiores do que 12 Informe que o valor não é válido.

ALGORITMO "ex18"

```
VAR
    num: INTEIRO
    mes: CARACTERE
INICIO
    ESCRIVA ("Informe um numero: ")
    LEIA (num)
    SE (num<1) OU (num>12) ENTAO
        ESCRIVAL ("Valor nao eh valido para mes")
    SENAO
        ESCOLHA (num)
        CASO 1
            mes <- "JANEIRO"
        CASO 2
            mes <- "FEVEREIRO"
        CASO 3
            mes <- "MARCO"
        CASO 4
            mes <- "ABRIL"
        CASO 5
            mes <- "MAIO"
        CASO 6
            mes <- "JUNHO"
        CASO 7
            mes <- "JULHO"
        CASO 8
            mes <- "AGOSTO"
        CASO 9
            mes <- "SETEMBRO"
        CASO 10
            mes <- "OUTUBRO"
        CASO 11
            mes <- "NOVEMBRO"
        CASO 12
            mes <- "DEZEMBRO"
        FIMESCOLHA
        ESCRIVAL ("Mes correspondente: ", mes)
    FIMSE
FIMALGORITMO
```


19. Uma empresa está aumentando o salário de seus empregados cfe. a categoria ocupada seguindo a tabela abaixo. Faça um algoritmo que leia o nome, a categoria e o salário atual de um funcionário e escreva o nome e o salário reajustado.

categoria	% de aumento sobre salário atual
A, C, F, H	10
B, D, E, G	15
I, K até R	20
J, S até Z	25

ALGORITMO "ex19"

```

VAR
    nome, categ: CARACTERE
    percentual: INTEIRO
    salario, salnovo: REAL
INICIO
    ESCREVA ("Informe nome, categoria e salario: ")
    LEIA (nome, categ, salario)
    ESCOLHA (categ)
        CASO "A", "C", "F", "H"
            percentual <- 10
        CASO "B", "D", "E", "G"
            percentual <- 15
        CASO "I", "K", "L", "M", "N", "O", "P", "Q", "R"
            percentual <- 20
        CASO "J", "S", "T", "U", "V", "X", "Z"
            percentual <- 25
    FIMESCOLHA
    salnovo <- salario * (1 + percentual/100)    // equivale: salario + perc/100 * salario
    ESCREVAL ("Nome", nome)
    ESCREVAL ("Salario novo R$", salnovo)
FIMALGOR

```

PENSAR EM REESTRUTURAR OS CASOS FINAIS

```

...
OUTROCASO:
SE (categ="I") OU ((categ>="K") E (categ<="R")) ENTÃO
    percentual <- 20
SENAO
    SE (categ="J") OU ((categ>="S") E (categ<="Z")) ...

```