

## RESOLUÇÕES LISTA REPITA-ATE (VisuAlg) – AULA 27SET22

3. Escreva um algoritmo que calcule a média dos números digitados pelo usuário, se eles forem pares. Termine a leitura se o usuário digitar zero (0).

```
algoritmo "ex3MediaPares"
var
    soma, cont, num: inteiro
inicio
    repita
        escreva("Informe valor (0 para sair): ")
        leia(num)
        se (num > 0) e (num MOD 2 = 0) entao
            soma <- soma + num // soma os pares
            cont <- cont + 1 // conta
        fimse
    ate (num = 0)
    se (cont > 0) então // nao pode ter divisao por 0
        escreval("Media dos numeros pares: ", soma / cont)
    senao
        escreval("Media dos numeros pares: 0")
    fimse
fimalgoritmo
```

4. Uma turma tem muitos alunos. Faça um algoritmo que:

- leia para cada aluno o seu nome e idade;
- escreva os nomes dos alunos que tem 18 anos;
- escreva a quantidade de alunos que tem idade acima de 20 anos.

A condição de parada é a resposta negativa do usuário à pergunta “Mais um aluno? S/N”.

```
algoritmo "ex4turma"
var
    nome, nomes18, resp: caracter
    idade, acima20: inteiro
inicio
    nomes18 <- "" // inicializa string vazia
    repita
        escreva("Nome: ")
        leia(nome)
        escreva("Idade: ")
        leia(idade)
        se (idade = 18) entao
            nomes18 <- nomes18 + nome + ", " // concatena nomes p/relatorio final
        senao
            se (idade > 20) entao
                acima20 <- acima20 + 1
            fimse
        fimse
    fimse
    escreva("Mais um aluno?S/N ")
    leia(resp)
    ate (resp = "N")
    escreval("Qte alunos acima de 20a: ", acima20)
    escreval("Alunos com 18 anos: ", nomes18)
fimalgoritmo
```

## RESOLUÇÕES LISTA ENQUANTO-FACA (VisuAlg) – AULA 27SET22

2. Uma máquina de biscoito está com problemas. Quando ligada, após 1 hora ela quebra 1 biscoito, na segunda hora ela quebra 3 biscoitos, na hora seguinte ela quebra 3 vezes a quantidade de biscoitos quebrados na hora anterior, e assim por diante. Faça um algoritmo que calcule quantos biscoitos são quebrados no final de cada dia (a máquina opera 16 horas por dia).

```
algoritmo "ex2biscoito"
var
    biscHora, hora, soma: inteiro
inicio
    biscHora <- 1
    hora <- 1
    enquanto (hora <= 16) faca
        escreval("Biscoitos perdidos nesta hora: ", biscHora)
        soma <- soma + biscHora
        hora <- hora + 1
        biscHora <- biscHora * 3
    fimenquanto
    escreval("Total de biscoitos perdidos: ", soma)
fimalgoritmo
```

4. Um determinado material radioativo perde metade de sua massa a cada 50 segundos. Dada a sua massa inicial em Kg, faça um algoritmo que determine o tempo necessário para que essa massa se torne menor que 0,5 gramas. Escreva a massa inicial, a massa final e o tempo.

```
algoritmo "ex4material"
var
    massaInicial, massaFinal: real
    segundos: inteiro
inicio
    escreva("Informe massa (em kg)")
    leia (massaInicial)
    massaFinal <- massaInicial
    enquanto (massaFinal > 0.005) faça
        massaFinal <- massaFinal / 2
        segundos <- segundos + 50
    fimenquanto
    escreval("Massa Inicial ", massaInicial, "g")
    escreval("Massa Final ", massaFinal, "g")
    escreval("Tempo ", segundos, "seg")
fimalgoritmo
```

## RESOLUÇÕES LISTA PARA-FACA (VisuAlg) – AULA 27SET22

2. Faça um algoritmo que calcule e escreva a soma da seguinte série de 100 termos:

$$S = 1 + 2 + 3 + 4 + 5 \dots + 100$$

```
algoritmo "ex2serie"
var
    num, soma: inteiro
inicio
    para num de 1 ate 100 faca
        soma <- soma + num
    fimpara
    escreval("Soma: ", soma)
fimalgoritmo
```

5. A série de Fibonacci é formada pela sequência 1, 1, 2, 3, 5, 8, 13, ...

Faça um algoritmo que escreva esta série até o n-ésimo termo. O valor n deve ser lido e deve ser maior do que 2 (use repita-ate para controlar a entrada de dados).

```
algoritmo "ex5fibonacci"
var
    termo1, termo2, cont, novotermino, n: inteiro
inicio
    termo1 <- 1
    termo2 <- 1
    cont <- 2
    repita
        escreva("Fibonacci - digite quantos termos deseja: ")
        leia(n)
    ate (n > 2)
    escreva("1, 1")
    para cont de 3 ate n faca
        novotermino <- termo1 + termo2
        escreva(", ", novotermino)
        termo1 <- termo2
        termo2 <- novotermino
    fimpara
fimalgoritmo
```