

RESOLUÇÕES LISTA REPITA-ATE (VisuAlg) – AULA 04OUT22

```
algoritmo "exerc5alturaturma"
var
    sexo, resp: caracter
    contTurma, contMulheres: inteiro
    altura, altTurma, altMulheres: real
inicio
    repita
        escreva("Altura: ")
        leia(altura)
        escreva("Sexo (F/M): ")
        leia(sexo)
        contTurma <- contTurma + 1
        altTurma <- altTurma + altura
        se (sexo = "F") entao
            contMulheres <- contMulheres + 1
            altMulheres <- altMulheres + altura
        fimse
        escreva("Mais um aluno(S/N)?")
        leia(resp)
    ate (resp = "N")
    escreval("Altura media da turma: ", altTurma/contTurma)
    se (contMulheres > 0) entao // nao pode ter divisao por 0
        escreval("Altura media das mulheres: ", altMulheres/contMulheres)
    senao
        escreval("Altura media das mulheres: 0")
    fimse
fimalgoritmo
```

RESOLUÇÕES LISTA REPITA-ATE (VisuAlg)

algoritmo "exerc6audiencia"

```
var
    resp: caracter
    canal, quant4, quant5, quant9, quant12, pessoas, total: inteiro
inicio
    quant4 <- 0    // pra lembrar de inicializar no C++
    quant5 <- 0
    quant9 <- 0
    quant12 <- 0
    repita
        repita
            escreva("Digite o canal da emissora: ")
            leia(canal)
        ate (canal=4 OU canal=5 OU canal=9 OU canal=12 OU canal=0)
        se (canal <> 0) então    // qdo tv ligada, se desligada nao se faz nada
            repita
                escreva("Quantas pessoas estão assistindo: ")
                leia(pessoas)
            ate(pessoas>0)
            escolha(canal)
            caso 4
                quant4 <- quant4 + pessoas
            caso 5
                quant5 <- quant5 + pessoas
            caso 9
                quant9 <- quant9 + pessoas
            caso 12
                quant12 <- quant12 + pessoas
            fimescolha
        fimse
        escreva("Mais uma casa(S/N)?")
        leia(resp)
    ate (resp = "N")
    total <- quant4 + quant5 + quant9 + quant12
    se (total > 0) então    // nao pode ter divisao por 0
        escreval("Canal 4 teve ", quant4*100.0/total, "% de audiencia")
        escreval("Canal 5 teve ", quant5*100.0/total, "% de audiencia")
        escreval("Canal 9 teve ", quant9*100.0/total, "% de audiencia")
        escreval("Canal 12 teve ", quant12*100.0/total, "% de audiencia")
    senao
        escreval("Todas as casas com tv desligada")
    fimse
fimalgoritmo
```

RESOLUÇÕES LISTA REPITA-ATE (VisuAlg)

```
algoritmo "exerc7vendedores"
var
    nome, op: caractere
    preco, total: real
    qtde: inteiro
inicio
    repita
        escreva("Nome do vendedor: ")
        leia(nome)
        total <- 0 // tem q zerar para cada vendedor, fica dentro do laco
        repita
            repita
                escreva("Preco unitario e qtde vendida do produto:")
                leia(preco, qtde)
            ate (preco>0) e (qtde>0)
            total <- total + preco * qtde
        repita
            escreva("Este funcionario tem mais uma venda? S/N")
            leia(op)
            ate (maiusc(op)="S") ou (maiusc(op)="N")
        ate (maiusc(op)="N")
        escreval("Vendedor:", nome)
        escreval("Total de vendas: R$", total)
        escreval("Salario: R$", total*0.3)
    repita
        escreva("Deseja digitar os dados de mais um vendedor? S/N")
        leia(op)
        ate (maiusc(op)="S") ou (maiusc(op)="N")
    ate (maiusc(op)="N")
fimalgoritmo
```

RESOLUÇÕES LISTA ENQUANTO-FACA (VisuAlg)

algoritmo "ex3divisores"

```
var
    numero, divisor, quant, cont, n: inteiro
inicio
    cont <- 0
    repita
        escreva("quantos valores voce deseja inserir? ")
        leia(n)
    ate (n > 0)
    enquanto (cont < n) faca
        escreva("Digite um numero inteiro: ")
        leia(numero)
        escreval("O numero lido eh: ", numero)
        escreva("Divisores: 1")      // lista de divisores, tem sempre o 1
        quant <- 1
        se (numero > 1) entao
            divisor <- 2
            enquanto (divisor <= numero DIV 2) faca
                se (numero MOD divisor = 0) entao
                    escreva(", ", divisor)
                    quant <- quant + 1
                fimse
                divisor <- divisor + 1
            fimenquanto
            escreva(", ", numero)
            quant <- quant + 1
        fimse
        escreval
        escreval("Qtde de divisores = ", quant)
        cont <- cont + 1
    fimenquanto
fim algoritmo
```

RESOLUÇÕES LISTA PARA-FACA (VisuAlg)

```
algoritmo "ex3seriereal"
```

```
var
```

```
    num: inteiro
```

```
    soma: real // em C++: lembrar de zerar
```

```
inicio
```

```
    para num de 1 ate 100 faca
```

```
        soma <- soma + 1/num // em C++: 1.0/num p/resultado float
```

```
    fimpara
```

```
    escreval("Soma: ", soma)
```

```
fimalgoritmo
```

```
algoritmo "ex4serierazao"
```

```
var
```

```
    num, i, denom, razao: inteiro
```

```
    soma: real // em C++: lembrar de zerar
```

```
inicio
```

```
    num <- 3
```

```
    denom <- 2
```

```
    razão <- 4
```

```
    para i de 1 ate 20 faca
```

```
        soma <- soma + num/denom // em C++: (float)num/denom p/resultado float
```

```
        num <- num + 2
```

```
        denom <- denom + razao
```

```
        razao <- razao + 2
```

```
    fimpara
```

```
    escreval("Soma: ", soma)
```

```
fimalgoritmo
```

RESOLUÇÕES LISTA PARA-FACA (VisuAlg)

algoritmo "ex7atletas"

```
// tratar 1o atleta antes do laço, para q ele seja referencia para comparacao
var
    inscr, altura, alturaMin, alturaMax, inscrMin, inscrMax, n, cont: inteiro
    media: real
inicio
    repita
        escreval("Qtde de atletas")
        leia(n)
    ate(n > 0)
    escreval("Primeiro atleta") // 1o atleta serve de comparacao inicial
    escreva("Digite a inscricao: ")
    leia(inscr)
    repita
        escreva("Digite a altura em centimetros: ")
        leia(altura)
    ate(altura > 0) E (altura < 300)
    alturaMin <- altura
    alturaMax <- altura
    inscrMin <- inscr
    inscrMax <- inscr
    media <- altura
    para cont de 2 ate n faca
        escreval("Proxima inscricao")
        escreva("Digite a inscricao: ")
        leia(inscr)
        repita
            escreva("Digite a altura em centímetros: ")
            leia(altura)
        ate(altura > 0) E (altura < 300)
        se (altura < alturaMin) entao
            alturaMin <- altura
            inscrMin <- inscr
        senao
            se (altura > alturaMax) entao
                alturaMax <- altura
                inscrMax <- inscr
        fimse
    fimse
    media <- media + altura
fimpara
escreva("Atleta ",inscrMin," eh o mais baixo - ",alturaMin," cm")
escreva("Atleta ",inscrMax," eh o mais alto - ",alturaMax," cm")
escreva("Altura media dos atletas: ", media/n," cm")
fimalgoritmo
```

RESOLUÇÕES LISTA PARA-FACA (VisuAlg)

```
algoritmo "ex8turmas"
var
    nAlunos, i, j, contPresentes: inteiro
    nota, total: real
    turma, matricula, freq: caractere
inicio
    para i de 1 ate 2 faca
        total <-0           // para cada turma tem q zerar estas 2 variaveis
        contPresentes <-0
        repita
            escreva("Identificacao da turma e qtde de alunos: ")
            leia(turma, nAlunos)
        ate(nAlunos > 0)
        para j de 1 ate nAlunos faca
            repita
                escreva("Matricula e frequencia na prova(A/P):")
                leia(matricula, freq)
            ate (maiusc(freq)="A") ou (maiusc(freq)="P")
            se maiusc(freq)="P" entao
                repita
                    escreva("Nota: ")
                    leia(nota)
                ate (nota>=0) e (nota<=10)
                contPresentes <- contPresentes + 1
                total <- total + nota
            fimse
        fimpara
        escreval("Turma: ", turma)
        se contPresentes>0 entao // nao pode ter divisao por 0
            escreval("Nota media dos presentes: ",total/contPresentes)
        fimse
        escreval("Ausencia: ",(nAlunos - contPresentes)*100/nAlunos,"%")
    fimpara
fimalgoritmo
```